This notebook is saved with all output deleted. To recreate all the calculations and the pictures go to the menu item

Evaluation-> Evaluate notebook (the shortcut is Alt v + o)

To evaluate an individual cells use Shift+Enter

When you are done, before saving the notebook, delete all output by menu item Cell->Delete all output (shortcut Alt c + l)

```
In[1]:= NotebookDirectory[]
Out[1]= C:\Dropbox\Work\myweb\Courses\Math_pages\Math_430\
```

```
In[2]:= NotebookFileName[]
Out[2]= C:\Dropbox\Work\myweb\Courses\Math_pages\Math_430\MoC_Burgers_eq1_v12.nb
```

Burgers’ Equation

The PDE

$$u(x,t)\frac{\partial u}{\partial x}(x,t)+\frac{\partial u}{\partial t}(x,t)=0$$

is called *Burgers’ equation*. This is NOT a linear equation. In this equation instead of the independent variable y we write t since it is convenient to think of it as time.

We will consider this equation subject to the initial condition

$u(x,0) = f(x)$ where $x \in \mathbb{R}$. (To make illustrations in Mathematica we will choose $f(x) = \text{Exp}[-x^2]$.

The vector field that we need for the characteristic equations of this equation is $< z, 1, 0 >$.

```
In[3]:= VPbe = {1.3`, -2.4`, 2.`}
Out[3]= {1.3, -2.4, 2.}
```

```
In[4]:= ChVecFiBE[{x_, t_, z_}] = {z, 1, 0}
Out[4]= {z, 1, 0}
```

In[5]:=
```
VPbe = {0.8937514594363056`, -2.8253081061674683`, 1.6336592159871852`}
```

Out[5]= {0.893751, -2.82531, 1.63366}

In[6]:=
```
vecsbe = VectorPlot3D[ChVecFiBE[{x, t, z}], {x, -2, 3}, {t, 0, 2},
  {z, -0, 1.5},
  VectorColorFunction → (RGBColor[0, 0.5, 0.5] &),
  VectorColorFunctionScaling → False,
  VectorStyle -> {Opacity[0.75], Thickness[0.006]},
  VectorPoints → {8, 12, 8} , VectorScale → {0.07, Scaled[0.6]},
  BoxRatios → {2, 2, 1}, PlotRange → {{-2, 2}, {0, 2}, {0, 1.5}},
  ImageSize → 500, ViewPoint → Dynamic[VPbe]]
```

Out[6]=



In[7]:=
```
VPbe
```

Out[7]= {0.893751, -2.82531, 1.63366}

Now we need to solve the initial value problem for the **characteristic equations**. For that we will specify the initial condition for the Burgers' equation. We choose u[x,0] = Exp[-x^2].

In[8]:=
```
Clear[solbe];
solbe[s_, ξ_] =
 FullSimplify[{x[s], t[s], z[s]} /.
   DSolve[{x'[s] == z[s], t'[s] == 1, z'[s] == 0, x[0] == ξ, t[0] == 0,
      z[0] == Exp[-ξ^2]}, {x[s], t[s], z[s]}, s][[1]]]
```

Out[9]= $\left\{e^{-\xi^2} s + \xi, s, e^{-\xi^2}\right\}$

The above triple, for a fixed ξ and for a varying s gives a curve in *xtz*-space. For many ξ-s we get many curves. These curves are the characteristics of Burgers' equation. However, for Burgers' equation the projected characteristics are more important. (Projected characteristics are the projections of the characteristics onto *xt*-plane.) Below we plot the projected characteristics. They are straight lines with slope $\text{Exp}[-\xi^2]$ in the xt-plane.

In[10]:=
```
ParametricPlot[Evaluate[Table[solbe[s, ξ][[{1, 2}]], {ξ, -9, 3, .05}]],
 {s, 0, 6}, PlotStyle → {Thickness[0.002]}, PlotRange → {{-2, 3}, {0, 2}},
 ImageSize → 600]
```

Out[10]=



Recall that the value of z along a fixed projected characteristic is constant. Thus at the points were projected characteristics intersect the function $u(x, t)$ should be having two different values. That is clearly impossible. Next we will try to answer the following question: What is the maximum time t_m for which no projected characteristics intersect below the line $t = t_m$. I will first guess that value, say $t_m = 1$.

In[11]:=
```
ParametricPlot[Evaluate[Table[solbe[s, ξ][[{1, 2}]], {ξ, -9, 3, .05}]],
  {s, 0, 6}, PlotStyle → {Thickness[0.002]},
  Epilog → {{Red, Line[{{-3, 1}, {5, 1}}]}}, PlotRange → {{-2, 3}, {0, 2}},
  ImageSize → 600]
```

Out[11]=



From this plot it is clear that $t_m > 1$. Next we will try to find the exact value of t_m.

First look at the surface that we found.

In[12]:=
```
solbe[s, ξ]
```

Out[12]= $\left\{e^{-\xi^2} s + \xi,\ s,\ e^{-\xi^2}\right\}$

This is a parametric equation of a surface in *xtz*-space.

In[13]:=
```
ParametricPlot3D[solbe[s, ξ], {ξ, -2, 3}, {s, 0, 2}, PlotPoints → {70, 30},
  PlotRange → {{-2, 3}, {0, 2}}, ImageSize → 600, AxesLabel → {"x", "t", "z"}]
```

Out[13]=



Think of a fixed time in the above plot, say $t = t_0$, and consider the curve $z = u(x, t_0)$. From the graph we can see that for small values of t_0 we have that $z = u(x, t_0)$ is a function. But for some larger values of t_0, say close to 2, we have that $z = u(x, t_0)$ is NOT a function.

Recall the equation of this surface:

In[14]:=
```
solbe[s, ξ]
```

Out[14]= $\left\{e^{-\xi^2}\, s + \xi,\ s,\ e^{-\xi^2}\right\}$

A lucky aspect of this equation is that the time is the second coordinate, that is the time equals s. Next I will explore the parametric curves with fixed $s = s_0$

In[15]:=
```
solbe[s0, ξ]
```

Out[15]= $\left\{e^{-\xi^2}\, s0 + \xi,\ s0,\ e^{-\xi^2}\right\}$

In[16]:=
```
Manipulate[ParametricPlot[solbe[s0, ξ][[{1, 3}]], {ξ, -2, 3},
  PlotPoints → 150, PlotRange → {{-2, 3}, {0, 1.5}}, ImageSize → 600],
 {s0, 0, 3}, ControlPlacement → Top]
```

Out[16]=



Since the parameter *s* is in fact the time, I will change the variable name to *t*. It does not make any difference mathematically but it might be easier to think about what is going on. I will also add some negative time to get the idea how this process evolves.

In[17]:=
```
Manipulate[ParametricPlot[solbe[t, ξ][[{1, 3}]], {ξ, -2, 3},
  PlotPoints → 150, PlotRange → {{-2, 3}, {0, 1.5}}, ImageSize → 600],
 {{t, 0}, -1, 3}, ControlPlacement → Top]
```

Out[17]=



As I pointed out earlier from the above graphs we can see that for small values of t we have that $z = u(x, t)$ is a function. But for some larger values of t, say close to 2, we have that $z = u(z, t)$ is NOT a function. The point is to find the exact value of the cut-off t. To find that t I will add the tangent vector to the above parametric curve. Recall that t is fixed and ξ is the variable, so the tangent vector is

In[18]:=
```
solbe[t, ξ][[{1, 3}]]
```

Out[18]= $\left\{e^{-\xi^2}\,t + \xi,\ e^{-\xi^2}\right\}$

In[19]:=
```
D[solbe[t, ξ][[{1, 3}]], ξ]
```

Out[19]= $\left\{1 - 2\,e^{-\xi^2}\,t\,\xi,\ -2\,e^{-\xi^2}\,\xi\right\}$

In[20]:=
```
Manipulate[ParametricPlot[solbe[t, ξ][[{1, 3}]], {ξ, -2, 3},
  PlotPoints → 150,
  Epilog →
   {{Arrow[{{e^(-ξ0^2) t + ξ0, e^(-ξ0^2)},
       {e^(-ξ0^2) t + ξ0, e^(-ξ0^2)} + {1 - 2 e^(-ξ0^2) t ξ0, -2 e^(-ξ0^2) ξ0}}]}},
  PlotRange → {{-2, 3}, {-1, 1.5}}, ImageSize → 600], {{t, 0}, -1, 3},
 {{ξ0, 0}, -2, 3}, ControlPlacement → Top]
```

Out[20]=



Based on this manipulation, can you calculate the the value of t for which $z = u(z, t)$ becomes multivalued?

Here is the calculation:

Calculate the tangent vector to the curve

In[21]:=
```
solbe[t, ξ][[{1, 3}]]
```

Out[21]= $\left\{e^{-\xi^2} t + \xi, e^{-\xi^2}\right\}$

for a fixed t

In[22]:=
```
D[solbe[t, ξ][[{1, 3}]], ξ]
```

Out[22]= $\left\{1 - 2 e^{-\xi^2} t \xi, -2 e^{-\xi^2} \xi\right\}$

The first component of the tangent vector is

In[23]:=
```
D[solbe[t, ξ][[{1, 3}]], ξ][[1]]
```

Out[23]= $1 - 2 e^{-\xi^2} t \xi$

Plot this function for a fixed t and then manipulate t:

In[24]:=
```
Manipulate[Plot[1 - 2 e^(-ξ^2) t ξ, {ξ, -1, 3}, PlotRange → {-0.2, 1.5}],
 {t, 0, 2}, ControlPlacement → Top]
```

Out[24]=


Find the derivative of the first component of the tangent vector.

In[25]:= FullSimplify$\left[D\left[1-2\, e^{-\xi^2}\, t\, \xi,\ \xi\right]\right]$

Out[25]= $2\, e^{-\xi^2}\, t\, \left(-1+2\, \xi^2\right)$

Find for which ξ the function $1-2\, e^{-\xi^2}\, t\, \xi$ takes a minimum.

In[26]:= Solve$\left[-2\, e^{-\xi^2}\, t+4\, e^{-\xi^2}\, t\, \xi^2 == 0,\ \xi\right]$

Out[26]= $\left\{\left\{\xi \to -\frac{1}{\sqrt{2}}\right\},\ \left\{\xi \to \frac{1}{\sqrt{2}}\right\}\right\}$

Calculate the second derivative of the first component to prove that it reaches the minimum at the above value of ξ:

In[27]:= FullSimplify$\left[D\left[1-2\, e^{-\xi^2}\, t\, \xi,\ \{\xi, 2\}\right]\ /.\ \left\{\xi \to \frac{1}{\sqrt{2}}\right\}\right]$

Out[27]= $4\, \sqrt{\frac{2}{e}}\, t$

Since t is positive, the last quantity is positive. Thus, the first component of the tangent vector has the minimum at $\xi = \frac{1}{\sqrt{2}}$

In[28]:= $\left(1-2\, e^{-\xi^2}\, t\, \xi\right)\ /.\ \left\{\xi \to \frac{1}{\sqrt{2}}\right\}$

Out[28]= $1-\sqrt{\frac{2}{e}}\, t$

Hence: at $\xi = 1/\sqrt{2}$ the derivative of the first component of the tangent vector takes the minimum value $1-\sqrt{\frac{2}{e}}\, t$. Finally, for which t is this minimum equal to 0:

In[29]:= Solve$\left[1-\sqrt{\frac{2}{e}}\, t == 0,\ t\right]$

Out[29]= $\left\{\left\{t \to \sqrt{\frac{e}{2}}\right\}\right\}$

In[30]:=
```
N[Sqrt[E / 2]]
```

Out[30]= 1.16582

Thus, the parametric equations

In[31]:=
```
solbe[t, ξ]
```

Out[31]= $\left\{e^{-\xi^2}\, t + \xi,\ t,\ e^{-\xi^2}\right\}$

represent a function for all (t,ξ) such that $t \in \left[0, \sqrt{e/2}\right]$ and $\xi \in \mathbb{R}$. Although we can not find an explicit formula for the function $u(x, t)$. For that formula we would need to solve

In[32]:=
```
Solve[Exp[-ξ^2] t + ξ == x, ξ]
```

Solve: This system cannot be solved with the methods available to Solve.

Out[32]= $\text{Solve}\left[e^{-\xi^2}\, t + \xi == x,\ \xi\right]$

In[33]:=
```
ParametricPlot3D[solbe[t, ξ], {ξ, -2, 3}, {t, 0, Sqrt[ⅇ/2]},
  PlotPoints → {70, 30}, PlotRange → {{-2, 3}, {0, Sqrt[ⅇ/2]}}, ImageSize → 600,
  AxesLabel → {"x", "t", "z"}]
```

Out[33]=


Mathematica algorithms can not solve this equation:

In[34]:=
```
Clear[ff, u];
DSolve[{u[x, t] × D[u[x, t], x] + D[u[x, t], t] == 0, u[x, 0] == ff[x]},
  u[x, t], {x, t}]
```

Out[34]=
```
DSolve[
  {u^(0,1)[x, t] + u[x, t] u^(1,0)[x, t] == 0, u[x, 0] == ff[x]}, u[x, t], {x, t}]
```

Or, with the specific initial condition that we used:

```
In[35]:= DSolve[{u[x, t] × D[u[x, t], x] + D[u[x, t], t] == 0, u[x, 0] == Exp[-x^2]},
   u[x, t], {x, t}]
```

Solve: Inverse functions are being used by Solve, so some solutions may not be found; use Reduce for complete solution information.

Out[35]= $\text{DSolve}\left[\left\{u^{(0,1)}[x,t] + u[x,t]\, u^{(1,0)}[x,t] == 0,\ u[x,0] == e^{-x^2}\right\},\ u[x,t],\ \{x,t\}\right]$